
Abstract. We present a proof in Agda that the substitutions in simply typed λ-calculus
form a category which has �nite products. We base our syntax on directly typed λ-calculus,
without de�ning �rst pure λ-calculus. We use parallel substitutions, a better point of view
to argue about substitutions. We show that we can reduce all the proofs to similar diagrams,
that make it automatic.

1 Introduction

The substitutions in simply-typed λ-calculus [1] form a category which has �nite products.
This has been proved with di�erent approaches (eg., see [2]). We propose here a new view,
with two particularities:

• we use a directly typed syntax, that is to say we consider only typed terms of λ-calculus;

• we prefer parallel substitution to syntactic substitution, because it is a better way to
abstract from substitutions and to make simpler proofs.

Parallel substitutions have been introduced by Abadi et al. [3] as a means to manipulate
substitutions as abstract objects. This is a better point of view to do proofs on it, than to use
the usual substitutions that work on syntactic rules.

The main idea of the proofs we present in this paper is to bring down to proofs for variables
as much as possible. Indeed, proofs over variables are far simpler than over terms. We intend to
explain how to bring terms down to variables, which is a mechanism that is quite automatic.
We notice that this way of doing leads to repetitive proofs that all match with only three
diagrams (see section 4.3 for more details).

In section 2, we will introduce the basic syntax of Agda. In section 3, we will present our
syntax for λ-calculus and substitutions. Section 4 will be devoted to prove the categorical
behaviour of substitutions.

2 A short introduction to Agda

The programs in this paper are terms in a dependent type theory. The development was made
in Agda, a dependently typed programming language with good support for programming

with inductively de�ned families of types [4]. It uses a Haskell-like syntax that allows type
dependence.

Here is a short introduction to Agda basic syntax. A complete tutorial is available on the
agda wiki [5].

2.1 Datatype declarations

Inductive datatypes are de�ned by a formation rule and constructors. For instance, one could
de�ne propositional equality as follows:

data _≡_ {A : Set } : A → A → Set where
r e f l : {a : A} → a ≡ a

For a given set A, the equality on this set is de�ned as a set of elements that are identical.
The use of the braces permits to de�ne arguments as implicit. The syntax _≡_ means that
this relation is in�x : we can write a ≡ a.

2.2 Function declarations

We can now de�ne a �rst function, which will consist in proving that the binary relation
de�ned above is symmetric:

sym : {A : Set } → {a b : A} → a ≡ b → b ≡ a
sym r e f l = r e f l

We �rst de�ne the type of the function sym, which is a dependent type : given a set A and
two elements a and b of A, if we have a proof that a ≡ b, then we can have a proof that b ≡ a.
The second line is the de�nition of the sym function. As A, a and b are implicit arguments, we
just have to provide a proof that a ≡ b (in order to obtain a proof that b ≡ a). We do this by
pattern matching on this proof. As the only constructor of the equality is re� , we have one
single case. The proof then reduces to the constructor re� .

We can prove the relation is transitive the same way:

t rans : {A : Set } → {a b c : A} → a ≡ b → b ≡ c → a ≡ c
t rans r e f l r e f l = r e f l

We now have the complete proof that our relation is an equivalence.

3 Explicit substitutions in simply typed λ-calculus

In this section, we present the simply typed λ-calculus with explicit substitutions. We use a
typed syntax, which is to say that we will de�ne inductively the typed terms of the λ-calculus,
instead of �rst de�ning terms and then introducing typing rules. In fact, we are only interested
in typed terms.

Substitutions in λ-calculus consist in replacing variables with terms inside a term. We can
see terms that are typed in context ∆ as trees whose leafs are variables in ∆. Applying a
substitution between Γ and ∆ to such terms is replacing those leafs with terms that are trees
whose leafs are variables in Γ. It is often done by applying some syntactic rules recursively,
but here, we prefer parallel substitutions: substitutions are just another means to form terms.

3.1 Syntax

3.1.1 λ-calculus

The set of types Ty : Set is de�ned with one single base type:

data Ty : Set where
base : Ty
_=>_ : Ty → Ty → Ty

To be able to type variables and terms, we need a set of contexts Context : Set, which are
backwards written lists of types:

data Context : Set where
empty : Context
ext : Context → Ty → Context

If Γ is a context and σ a type, the set of the variables that have type σ in context Γ
Var Γ σ : Set is de�ned as follows:

data Var : Context → Ty → Set where
v l a s t : f o r a l l {Γ τ } → Var ( ext Γ τ ) τ
weak : f o r a l l {σ Γ τ } → Var Γ τ → Var ( ext Γ σ ) τ

And so is the set of the terms that have type σ in context Γ Term Γ σ : Set:

data Term : Context → Ty → Set where
var : f o r a l l {Γ σ} → Var Γ σ → Term Γ σ
lam : f o r a l l {Γ τ σ} → Term ( ext Γ τ ) σ → Term Γ (τ => σ )
app : f o r a l l {Γ τ σ} → Term Γ (τ => σ ) → Term Γ τ → Term Γ σ

We notice that both Var and Term have the same signature Context → Ty → Set. As a result,
variables and terms have lots of similar de�nitions, and we can abstract from this. If T :

Context → Ty → Set, we will be able to de�ne substitutions that replace variables with elements
of type T provided three functions exist:

• vr : forall {Γ σ} → Var Γ σ → T Γ σ

• tm : forall {Γ σ} → T Γ σ → Term Γ σ

• wk : forall {Γ σ τ} → T Γ σ → T (ext Γ τ) σ

To compose substitutions, we will need a fourth function:

• subst : forall {Γ ∆ σ} → T ∆ σ → Subst T Γ ∆ → T Γ σ

These functions can form a kit, as suggested in [6], that we will instantiate for variables
and for terms. To represent this kit, we use record types:

r ecord SubstKit (T : Context → Ty → Set ) : Set where
f i e l d

vr : f o r a l l {Γ σ} → Var Γ σ → T Γ σ
tm : f o r a l l {Γ σ} → T Γ σ → Term Γ σ
wk : f o r a l l {Γ σ τ } → T Γ σ → T ( ext Γ τ ) σ

r ecord SubstKit+ (T : Context → Ty → Set ) : Set where
f i e l d

k i t : SubstKit T
subst : f o r a l l {Γ ∆ σ} → T ∆ σ → Subst T Γ ∆ → T Γ σ

3.1.2 Substitutions

Given T : Context → Ty → Set and two contexts Γ and ∆, we can de�ne the set of substitutions
that transforms elements of T ∆ into elements of T Γ as follows:

data Subst (T : Context → Ty → Set ) : Context → Context → Set where
substEmpty : f o r a l l {Γ} → Subst T Γ empty
_,_ : f o r a l l {Γ ∆ σ} → Subst T Γ ∆ → T Γ σ → Subst T Γ ( ext ∆ σ )

3.1.3 Categorical combinators

The categorical combinators consist in weakening and lifting substitutions, the identity sub-
stitution, and the composition of substitutions.

Weakening a substitution consists in extending the codomain:

_+_ : f o r a l l {σ T Γ ∆} → Subst T Γ ∆ → SubstKit T → Subst T ( ext Γ σ ) ∆
substEmpty + _ = substEmpty
(u , t ) + k = (u + k ) , ( ( SubstKit .wk k ) t )

Lifting a substitution consists in extending both the domain and the codomain:

_++_ : f o r a l l {σ T Γ ∆} → Subst T Γ ∆ → SubstKit T →
Subst T ( ext Γ σ ) ( ext ∆ σ )

u ++ k = (u + k ) , ( ( SubstKit . vr k ) v l a s t )

We can now de�ne the identity substitutions. We �rst de�ne an identity substitution for
variables, then lift it using the constructor var to obtain the identity substitution for terms.

idSVar : {Γ : Context} → Subst Var Γ Γ
idSVar {empty} = substEmpty
idSVar { ext Γ σ} = idSVar ++ vk

substTermOfSubstVar : f o r a l l {Γ ∆} → Subst Var Γ ∆ → Subst Term Γ ∆
substTermOfSubstVar substEmpty = substEmpty
substTermOfSubstVar ( s , v ) = ( substTermOfSubstVar s ) , ( var v )

idSTerm : {Γ : Context} → Subst Term Γ Γ
idSTerm = substTermOfSubstVar idSVar

The composition of two substitutions of the same signature can be de�ned as follows:

compS : f o r a l l {T Γ ∆ Θ} → SubstKit+ T → Subst T Γ ∆ → Subst T ∆ Θ →
Subst T Γ Θ

compS _ _ substEmpty = substEmpty
compS k+ s ( s ' , t ) = (compS k+ s s ' ) , ( ( SubstKit+. subst k+) t s )

3.2 Substitution functions

3.2.1 Variables

If T : Context → Ty → Set, we can apply a Subst T to any variable, and then obtain an element of
type T:

substVar : f o r a l l {T Γ ∆ σ} → Subst T Γ ∆ → Var ∆ σ → T Γ σ
substVar substEmpty ( )
substVar ( s , t ) v l a s t = t
substVar ( s , t ) (weak v ) = substVar s v

_[_] : f o r a l l {T Γ ∆ σ} → Var ∆ σ → Subst T Γ ∆ → T Γ σ
v [ s ] = substVar s v

We have now all the tools to de�ne kits for variables:

vk : SubstKit Var
vk = record

{ vr = (\ a → a )
; tm = var
; wk = weak
}

vk+ : SubstKit+ Var
vk+ = record

{ k i t = vk
; subst = _[_]
}

3.2.2 Terms

If T : Context → Ty → Set, we can apply a Subst T to any term, and then obtain a term:

substTerm : f o r a l l {T Γ ∆ σ} → SubstKit T → Subst T Γ ∆ → Term ∆ σ →
Term Γ σ

substTerm k s ( var v ) = SubstKit . tm k (v [ s ] )
substTerm k s ( lam t ) = lam ( substTerm k ( s ++ k ) t )
substTerm k s ( app t1 t2 ) = app ( substTerm k s t1 ) ( substTerm k s t2 )

We will need to weaken terms to complete our kits. Weakening a variable is just applying
the constructor weak; but to weaken a term t, we substitute the weakened idSVar to t (this will
allow us to bring proofs for terms down to proofs for variables):

termWeak : f o r a l l {τ Γ σ} → Term Γ σ → Term ( ext Γ τ ) σ
termWeak t = substTerm vk ( idSVar + vk ) t

We can now de�ne kits and more pleasant notations:

tk : SubstKit Term
tk = record

{ vr = var
; tm = (\ a → a )
; wk = termWeak
}

_[_] 1 : f o r a l l {Γ ∆ σ} → Term ∆ σ → Subst Var Γ ∆ → Term Γ σ
t [ u ] 1 = substTerm vk u t

_[_] 2 : f o r a l l {Γ ∆ σ} → Term ∆ σ → Subst Term Γ ∆ → Term Γ σ
t [ u ] 2 = substTerm tk u t

tk+ : SubstKit+ Term
tk+ = record

{ k i t = tk
; subst = _[_] 2
}

We can also have corresponding notations for composition:

_◦1_ : f o r a l l {Γ ∆ Θ} → Subst Var Γ ∆ → Subst Var ∆ Θ → Subst Var Γ Θ
s ◦1 s ' = compS vk+ s s '

_◦2_ : f o r a l l {Γ ∆ Θ} → Subst Term Γ ∆ → Subst Term ∆ Θ → Subst Term Γ
Θ

s ◦2 s ' = compS tk+ s s '

3.3 Extra functions

We will also need to compose substitutions of di�erent signatures, ie variable and term sub-
stitutions, in both senses:

_◦3_ : f o r a l l {Γ ∆ Θ} → Subst Var Γ ∆ → Subst Term ∆ Θ → Subst Term Γ Θ
_ ◦3 substEmpty = substEmpty
s ◦3 ( s ' , t ) = ( s ◦3 s ' ) , ( t [ s ] 1 )

_◦4_ : f o r a l l {Γ ∆ Θ} → Subst Term Γ ∆ → Subst Var ∆ Θ → Subst Term Γ Θ
_ ◦4 substEmpty = substEmpty
s ◦4 ( s ' , v ) = ( s ◦4 s ' ) , ( v [ s ] )

4 The substitutions form a category which has �nite products

In this section, we intend to prove that the structure we de�ned in section 3 is a category with
�nite products. We will �rst prove it for variable substitutions, then for term substitutions,
by bringing down to variable cases. But �rst, we need a few preliminary lemmas.

All the proofs have been checked with Agda. The complete proof is available online [7].

4.1 Preliminary lemmas

Five lemmas are used to prove the compatibility of the structural equality with constructors
such as weak, var, lam, app and substExt. They are respectively named re�Weak, re�Var, re�Lam,
re�App and re�SubstExt. They are as easy to prove as sym and trans, and have the following
prototype:

r e f lCons : f o r a l l {p1.1 p1.2 . . . pn.1 pn.2} −>
p1.1 ≡ p1.2 −> . . . −> pn.1 ≡ pn.2 −>
Cons p1.1 . . . pn.1 ≡ Cons p1.2 . . . pn.2

We will also need to prove that, if s1 and s2 are two structurally equal substitutions, and
t1 and t2 are two structurally equal terms or variables, then t1[s1] ≡ t2[s2]. This drives to
three lemmas that can be proved like the former ones:

r e f l S ub s t : f o r a l l {∆ σ} → { t1 t2 : Term ∆ σ} → f o r a l l {Γ} →
{ s1 s2 : Subst Term Γ ∆} → s1 ≡ s2 → t1 ≡ t2 →

t1 [ s1 ] 2 ≡ t2 [ s2 ] 2

r e f l Sub s t 2 : f o r a l l {∆ σ} → { t1 t2 : Var ∆ σ} → f o r a l l {T Γ} →
{ s1 s2 : Subst T Γ ∆} → s1 ≡ s2 → t1 ≡ t2 →

t1 [ s1 ] ≡ t2 [ s2 ]

r e f l Sub s t 3 : f o r a l l {∆ σ} → { t1 t2 : Term ∆ σ} → f o r a l l {Γ} →
{ s1 s2 : Subst Var Γ ∆} → s1 ≡ s2 → t1 ≡ t2 →

t1 [ s1 ] 1 ≡ t2 [ s2 ] 1

4.2 Proofs for variables

Here we will present the main ideas to prove that substitutions for variables form a category
with �nite products.

4.2.1 The identity is neutral

We have to show that the identity is neutral for the composition of substitutions, both on the
left and on the right. The proofs will be completely di�erent, as the de�nition of composition
is non symmetric on the �rst and on the second substitutions.

The identity is neutral on the left We �rst need to prove this property:

Proposition 1 If v is a variable, s a substitution and σ a type, then:

v[s+σ] ≡ (v[s])+σ

This is proved by simple pattern matching on v:

eqWeak : f o r a l l {Γ ∆ σ τ } → ( s : Subst Var Γ ∆) → ( v : Var ∆ σ ) →
v [ _+_ {τ } s vk ] ≡ weak (v [ s ] )

eqWeak substEmpty ( )

eqWeak ( s , v1 ) v l a s t = r e f l
eqWeak ( s , v1 ) (weak v2 ) = eqWeak s v2

We can then mutually prove that:

Proposition 2 Given v a variable: {
v[id] ≡ v
v[id+σ] ≡ v+σ

Once more, pattern matching makes the proof really simple:

nLVar : f o r a l l {Γ σ} → ( v : Var Γ σ ) → v [ idSVar ] ≡ v
nLVar v l a s t = r e f l
nLVar (weak v ) = eqId v

eqId : f o r a l l {Γ σ τ } → ( v : Var Γ σ ) → v [ _+_ {τ } idSVar vk ] ≡
weak v

eqId v = trans (eqWeak idSVar v ) ( ref lWeak (nLVar v ) )

This leads automatically to the main theorem:

Theorem 1 If s is a variable substitution:

id ◦ s ≡ s

neutralLVar : f o r a l l {Γ ∆} → ( s : Subst Var Γ ∆) → idSVar ◦1 s ≡ s
neutralLVar substEmpty = r e f l
neutralLVar ( s , v ) = re f l Subs tExt ( neutralLVar s ) (nLVar v )

The identity is neutral on the right We need one single lemma:

Proposition 3 Given two substitutions s and s′ and a variable v:

(s, v) ◦ s′+σ ≡ s ◦ s′

nRVar : f o r a l l {Γ ∆ Θ σ} → ( s : Subst Var Γ ∆) → ( s ' : Subst Var ∆ Θ) →
( v : Var Γ σ ) → ( s , v ) ◦1 ( s ' + vk ) ≡ s ◦1 s '

nRVar _ substEmpty _ = r e f l
nRVar s ( s ' , _) v = re f l Subs tExt (nRVar s s ' v ) r e f l

Then we have our main theorem:

Theorem 2 If s is a variable substitution:

s ◦ id ≡ s

neutralRVar : f o r a l l {Γ ∆} → ( s : Subst Var Γ ∆) → s ◦1 idSVar ≡ s
neutralRVar substEmpty = r e f l
neutralRVar ( s , v ) = re f l Subs tExt ( t rans (nRVar s idSVar v ) ( neutralRVar

s ) ) r e f l

4.2.2 The composition of substitutions is associative

We just have to previously prove that applying the composition of two substitutions is the
same as applying the �rst one, then the second one:

Proposition 4 s and s′ are two substitutions, v is a variable.

v[s ◦ s′] ≡ (v[s′])[s]

We prove it by pattern matching on s′ and v:

aCSVar : f o r a l l {Γ ∆ Θ σ} → (u : Subst Var Γ ∆) → ( v : Subst Var ∆ Θ) →
(v ' : Var Θ σ ) → v ' [ u ◦1 v ] ≡ (v ' [ v ] ) [ u ]

aCSVar _ substEmpty ( )
aCSVar _ (_ , _) v l a s t = r e f l
aCSVar u (v , _) (weak v ' ) = aCSVar u v v '

The theorem is now provable:

Theorem 3 u, v and w are three variable substitutions.

(u ◦ v) ◦ w ≡ u ◦ (v ◦ w)

assoCompSVar : f o r a l l {Γ ∆ Θ Ξ} → (u : Subst Var Γ ∆) →
( v : Subst Var ∆ Θ) → (w : Subst Var Θ Ξ) →

(u ◦1 v ) ◦1 w ≡ u ◦1 ( v ◦1 w)
assoCompSVar _ _ substEmpty = r e f l
assoCompSVar u v (w , t ) = re f l Subs tExt ( assoCompSVar u v w) (aCSVar u v t )

4.2.3 This category has �nite products

The constructor _,_ has prototype Subst Var Γ ∆ −> Var Γ σ −> Subst Var Γ (ext ∆ σ). By de-curryfy-
ing this function, we obtain a function of prototype Subst Var Γ ∆ × Var Γ σ −> Subst Var Γ (ext ∆ σ).

We can consider the product Subst Var Γ ∆ × Var Γ σ as a product in the category theoretic
sense. To do so, we need to �nd two functions π1 and π2 that satisfy the following properties:
for all substitution u and variable v,

• π1(u, v) = u

• π2(u, v) = v

• (π1(u), π2(u)) = u

We propose:

• π1(u) = id+σ ◦ u

• π2(u) = vlast[u]

We can now prove the required properties:

Theorem 4 π1(u, v) = u

This is simply done by using previously proved theorems:

π1 : f o r a l l {Γ ∆ σ} → ( s : Subst Var Γ ∆) → ( v : Var Γ σ ) →
( s , v ) ◦1 ( idSVar + vk ) ≡ s

π1 s v = trans (nRVar s idSVar v ) ( neutralRVar s )

Theorem 5 π2(u, v) = v

This is pure re�exivity:

π2 : f o r a l l {Γ ∆ σ} → ( s : Subst Var Γ ∆) → ( v : Var Γ σ ) →
v l a s t [ s , v ] ≡ v

π2 _ _ = r e f l

Applying π2 makes sense only if its argument is an extended substitution. The third
theorem then is the following one:

Theorem 6 (π1(u, v), π2(u, v)) = (u, v)

Here is the proof in Agda:

sp : f o r a l l {Γ ∆ σ} → ( s : Subst Var Γ ∆) → ( v : Var Γ σ ) →
( ( s , v ) ◦1 ( idSVar + vk ) ) , ( v l a s t [ s , v ] ) ≡ s , v

sp s v = re f l Subs tExt (π1 s v ) (π2 s v )

4.3 Proofs for terms

Here we will present the main ideas to prove that substitutions for terms form a category
with �nite products. The main idea of all the proofs we will present is to bring down to
variable cases, on which proofs are simpler. For instance, to prove that the composition of
term substitutions is associative, we �rst study composition of a term substitution and a
variable one.

There are three main ways of conducting proofs:

• by pattern matching on variables: these are similar to the proofs presented in section
4.2;

• by pattern matching on terms. Given a substitution u, a term t, and two expressions f
and g that contain u and t, the outline of the proof is the following one:

proo f : f o r a l l {Γ ∆ σ} −> (u : Subst Term Γ ∆) −> ( t : Term ∆ σ ) −>
f u t ≡ g u t

proo f substEmpty ( var ( ) )
proo f (_ , _) ( var v l a s t ) = r e f l
p roo f (u , _) ( var (weak v ) ) = proo f u ( var v )

proo f u ( lam t ) = ref lLam ( t rans ( proo f (u ++ tk ) t ) . . . )

proo f u ( app t1 t2 ) = re f lApp ( proo f u t1 ) ( proo f u t2 )

• by pattern matching on substitutions. Given a substitution u and two expressions f and
g that contain u, the outline of the proof is the following one:

proo f2 : f o r a l l {Γ ∆} −> (u : Subst Term Γ ∆) −> f u ≡ g u
proo f2 substEmpty = r e f l
proo f2 (u , t ) = re f l Subs tExt ( proo f2 u) . . .

The proofs for terms may be really repetitive. In the following sections, we will present
the required lemmas and explain which kind of proof described above they refer to. One can
check the complete proof to �nd more details.

4.3.1 The identity is neutral

We have to show that the identity is neutral for the composition of substitutions, both on the
left and on the right. Once more, the proofs will be completely di�erent, as the de�nition of
composition is non symmetric on the �rst and on the second substitutions.

The identity is neutral on the left De�ning the identity substitution for terms as a
lifting of the identity substitution for variables with the constructor var, we can use proofs for
variables to prove that the identity substitution for terms is neutral on the left.

We �rst have to prove that _+_ and substTermOfSubstVar commute:

substWeakExchange : f o r a l l {Γ ∆ σ} → ( s : Subst Var Γ ∆) →
_+_ {σ} ( substTermOfSubstVar s ) tk ≡ substTermOfSubstVar ( s + vk )

This is proved by pattern matching on the substitution and using the lemma eqId.
We also want to prove that _[_] and substTermOfSubstVar commute:

substTermAndVar : f o r a l l {Γ ∆ σ} → ( s : Subst Var Γ ∆) → ( t : Term ∆ σ )
→ t [ substTermOfSubstVar s ] 2 ≡ t [ s ] 1

This is done by pattern matching on the term.
As for variables, we have to prove that applying the identity substitution to a term does

not change this term:

Proposition 5 t is a term.

t[id] ≡ t

nLTerm2 : f o r a l l {Γ σ} → ( t : Term Γ σ ) → t [ idSVar ] 1 ≡ t
nLTerm2 ( var v ) = r e f lVa r (nLVar v )
nLTerm2 ( lam t ) = ref lLam (nLTerm2 t )
nLTerm2 ( app t1 t2 ) = re f lApp (nLTerm2 t1 ) (nLTerm2 t2 )

nLTerm : f o r a l l {Γ σ} → ( t : Term Γ σ ) → t [ idSTerm ] 2 ≡ t
nLTerm t = trans ( substTermAndVar idSVar t ) (nLTerm2 t )

We can now have our main theorem:

Theorem 7 u is a term substitution.

id ◦ u ≡ u

neutralLTerm : f o r a l l {Γ ∆} → (u : Subst Term Γ ∆) → idSTerm ◦2 u ≡ u

The proof is exactly the same as for variables.

The identity is neutral on the right The proof is exactly the same as for variables. We
will only provide the prototypes of the functions.

Theorem 8 u is a term substitution.

u ◦ id ≡ u

nRTerm : f o r a l l {Γ ∆ Θ σ} → (u : Subst Term Γ ∆) → ( s : Subst Var ∆ Θ)
→ ( t : Term Γ σ ) → (u , t ) ◦2 ( substTermOfSubstVar ( s + vk ) ) ≡

u ◦2 ( substTermOfSubstVar s )

neutralRTerm : f o r a l l {Γ ∆} → (u : Subst Term Γ ∆) → u ◦2 idSTerm ≡ u

4.3.2 The composition of substitutions is associative

To do this proof, we will bring down to proofs on variables. As a result, for each intermediate
result, there will be a series of lemmas, that demonstrate the same result, but once for variables,
once for terms, and sometimes mixing terms and variables.

We will need three series of lemmas, that can be deduced from one another. Each lemma
deals with substitutions over terms and variables, in such a way that lemmas for terms are
deduced from lemmas for variables and lemmas for variables are easily provable.

These three series are:

• aCSTerm: applying the composition of two (variable or term) substitutions to a term is
like applying the �rst one then the second one (really similar to aCSVar):

Proposition 6 t is a term or a variable, u and v are substitutions.

t[u ◦ v] ≡ (t[v])[u]

• eqWeakTerm, that describes the behaviour of weakening a substituted term (really similar
to eqWeak):

Proposition 7 Given t a term or a variable, u a substitution and σ a type:

(t[u])+σ ≡ t+σ[u++σ]

• compWeak: lifting the composition of two (variable or term) substitutions is structurally
equal to composing the two lifted substitutions:

Proposition 8 u and v are two substitutions, σ is a type.{
(u ◦ v)+σ ≡ u++σ ◦ v+σ

(u ◦ v)++σ ≡ u++σ ◦ v++σ

aCSTerm We �rst have to prove that the weakened identity just weakens substitutions, for
both variables and terms:

Proposition 9 Given u a (variable or term) substitution, and σ a type:{
id+σ ◦ u ≡ u+σ

u+σ ≡ u++σ ◦ id+σ

This is proved by pattern matching on the substitution u, and corresponds in the Agda
program to the lemmas weakIn.

Transitivity automatically leads to this property:

Proposition 10 Given u a (variable or term) substitution, and σ a type:

id+σ ◦ u ≡ u++σ ◦ id+σ

We can now prove our main property, it is to say that applying the composition of two
substitutions to a term is like applying the �rst one then the second one, for each kind of
substitutions (Proposition 6).

This is done by pattern matching on term t and using the series of lemmas called compWeak.
Here is one example in Agda:

aCSTerm : f o r a l l {Γ ∆ Θ σ} → (u : Subst Term Γ ∆) →
( v : Subst Term ∆ Θ) → ( t : Term Θ σ ) →
t [ u ◦2 v ] 2 ≡ ( t [ v ] 2 ) [ u ] 2

aCSTerm u substEmpty ( var ( ) )
aCSTerm u ( s , t ) ( var v l a s t ) = r e f l
aCSTerm u ( s , t ) ( var (weak v ) ) = aCSTerm u s ( var v )

12



aCSTerm u s ( lam t ) = ref lLam
( t rans ( r e f l S ub s t {_} {_} { t } { t } (compWeak u s )

r e f l )
(aCSTerm (u ++ tk ) ( s ++ tk ) t ) )

aCSTerm u s ( app t1 t2 ) = re f lApp (aCSTerm u s t1 ) (aCSTerm u s t2 )

eqWeakTerm Using symmetry, transitivity and compatibility with substitution, we can obtain
this property (Proposition 7) thanks to the series of lemmas aCSTerm.

Proof
(t[u])+σ ≡ (t[u])[id+σ] by de�nition

≡ t[id+σ ◦ u] by Proposition 6

≡ t[u++σ ◦ id+σ] by Proposition 10

≡ (t[id+σ])[u++σ] by Proposition 6

≡ t+σ[u++σ] by de�nition

�

compWeak We want to prove that lifting the composition of two (variable or term) substitu-
tions is structurally equal to composing the two lifted substitutions (Proposition 8).

The second property automatically comes from the �rst one (it consists in applying the def-
inition of _++_) and the �rst property can be proved by pattern matching on the substitution
v and using eqWeakTerm. Here is one example in Agda:

compWeakAux : f o r a l l {Γ ∆ Θ σ} → (u : Subst Term Γ ∆) →
( v : Subst Term ∆ Θ) →
_+_ {σ} (u ◦2 v ) tk ≡ (u ++ tk ) ◦2 ( v + tk )

compWeakAux u substEmpty = r e f l
compWeakAux u (v , t ) = re f l Subs tExt (compWeakAux u v ) (eqWeakTerm u t )

compWeak : f o r a l l {Γ ∆ Θ σ} → (u : Subst Term Γ ∆) →
( v : Subst Term ∆ Θ) →
_++_ {σ} (u ◦2 v ) tk ≡ (u ++ tk ) ◦2 ( v ++ tk )

compWeak u v = re f lSubs tExt (compWeakAux u v ) r e f l

Main theorem Thanks to the lemma aCSTerm applied to term substitutions, we can prove
our main theorem exactly the same way we proved it for variables:

Theorem 9

(u ◦ v) ◦ w ≡ u ◦ (v ◦ w)

assoCompSTerm : f o r a l l {Γ ∆ Θ Ξ} → (u : Subst Term Γ ∆) →
( v : Subst Term ∆ Θ) → (w : Subst Term Θ Ξ) →
(u ◦2 v ) ◦2 w ≡ u ◦2 ( v ◦2 w)

assoCompSTerm u v substEmpty = r e f l
assoCompSTerm u v (w , t ) = re f l Subs tExt (assoCompSTerm u v w) (aCSTerm u v

t )

13



4.3.3 This category has �nite products

We can observe the same properties as for variables concerning �nite products. The theorems
required to prove that this category has �nite products are as simple to prove as for variables,
except for the �rst projector. As he involves the identity substitution for terms, we have to
bring it down to variables, as in section 4.3.1.

First projector We will have to �rst prove that the composition of an extended substi-
tution and the weakened identity is the substitution. We can do this by mutual recursion,
demonstrating these two facts:

• the identity substitution for variables is neutral on the right when composing with a
term substitution;

• the composition of an extended substitution and a weakened substitution is the compo-
sition of the two main substitutions.

The proof relies on pattern matching on substitutions:

neutralRTermVar : f o r a l l {Γ ∆} → (u : Subst Term Γ ∆) → u ◦4 idSVar ≡ u
neutralRTermVar substEmpty = r e f l
neutralRTermVar (u , t ) = re f l Subs tExt ( extWeakId u t ) r e f l

extWeak : f o r a l l {Γ ∆ Θ σ} → (u : Subst Term Γ ∆) →
( s : Subst Var ∆ Θ) → ( t : Term Γ σ ) →
(u , t ) ◦4 ( s + vk ) ≡ u ◦4 s

extWeak _ substEmpty _ = r e f l
extWeak u ( s , _) t = re f l Subs tExt ( extWeak u s t ) r e f l

extWeakId : f o r a l l {Γ ∆ σ} → (u : Subst Term Γ ∆) → ( t : Term Γ σ ) →
(u , t ) ◦4 ( idSVar + vk ) ≡ u

extWeakId u t = trans ( extWeak u idSVar t ) ( neutralRTermVar u)

weakExtTerm : f o r a l l {Γ ∆ Θ σ} → (u : Subst Term Γ ∆) →
( v : Subst Term ∆ Θ) → ( t : Term Γ σ ) →
(u , t ) ◦2 ( v + tk ) ≡ u ◦2 v

weakExtTerm _ substEmpty _ = r e f l
weakExtTerm u (v , t ' ) t = re f l Subs tExt (weakExtTerm u v t )

( t rans (sym (aCSTerm2 (u , t ) ( idSVar +

vk ) t ' ) )
( r e f l S ub s t {_} {_} {t ' } {t ' } ( extWeakId

u t ) r e f l ) )

We can now prove our main theorem:

Theorem 10 π1(u, v) = u

π '1 : f o r a l l {Γ ∆ σ} → (u : Subst Term Γ ∆) → ( t : Term Γ σ ) →
(u , t ) ◦2 ( idSTerm + tk ) ≡ u

π '1 u t = trans (weakExtTerm u idSTerm t ) ( neutralRTerm u)

Second projector

Theorem 11 π2(u, v) = v

This is pure re�exivity:

π '2 : f o r a l l {Γ ∆ σ} → (u : Subst Term Γ ∆) → ( t : Term Γ σ ) →
( var v l a s t ) [ u , t ] 2 ≡ t

π '2 _ _ = r e f l

Surjective pairing Applying π2 makes sense only if its argument is an extended substitu-
tion. The third theorem then is the following one:

Theorem 12 (π1(u, v), π2(u, v)) = (u, v)

Here is the proof in Agda:

sp ' : f o r a l l {Γ ∆ σ} → (u : Subst Term Γ ∆) → ( t : Term Γ σ ) →
( ( u , t ) ◦2 ( idSTerm + tk ) ) , ( ( var v l a s t ) [ u , t ] 2 ) ≡ u , t

sp ' u t = re f l Subs tExt (π '1 u t ) (π '2 u t )

5 Conclusion

We have a proof of the substitutions in simply-typed λ-calculus forming a category which has
�nite products. This proof has been completely checked using the proof assistant Agda.

This result is not new, but the approach is interesting.
First of all, we used a directly typed syntax. We leave non typable terms aside, focusing

only on what is interesting for our proofs. This leads us to simpler proofs, that are structurally
well-organized and automatic.

We stress the fact that de�ning functions for terms from functions for variables increases
considerably this automation for proofs. We de�ne the identity substitution for terms and
the way to weaken terms using the identity substitution for variables. As a result, a proof for
terms reduces easily to a series of proofs melting terms and variables, and �nally to proofs for
variables.

Instead of implicit substitutions that are de�ned recursively on the structure of terms,
we prefer parallel substitutions. This allows a more abstract point of view than pointwise
substitutions, as it can be seen as a means like another to form typed terms in a context.

We believe this work can lead to a kind of an automation for proofs over simply-typed
λ-calculus in Agda. We have to justify this assertion by extending it to many other proofs. At
least we produced a reliable Agda code that forms a base to conduce similar demonstrations.

